
ABSTRACT 
In programming activities during clinical trial reporting, situations can arise when a change in a portion of SAS® code 
is warranted after the programs are finalized and moved into a production environment.  It is a very common situation 
during the clinical trial reporting process. There may be one minor change in the SAS® code to be applied repeatedly 
to a large number of programs. One way to achieve this is by changing the hard code to each of the programs. 
However, if the number of programs is sufficiently large, then the risk of making mistakes due to hard coding 
increases. Therefore, it is always better to avoid repeated hard coding, instead it should be coded only once and the 
SAS® macro tool should automatically find that string of code and go into each of the programs and make the 
necessary changes. That way the error due to hard coding is minimized.    
 
This SAS® macro program tool first identifies the number of programs residing in a certain library and then goes to 
each program one by one, identifies the lines or string of text to be modified and finally makes necessary changes. 
Then it automatically saves the modified program in another output library.  This SAS® macro tool is created for the 
PC platform of Microsoft Windows and also for MVS Mainframe. This macro minimizes risk of errors due to typing. 
Furthermore, it could be applied to a large number of programs in need of modification. 

INTRODUCTION 
Changing SAS® code is a frequent process in programming and reporting activities. When dealing with a large 
number of reports that need changes, copying and pasting code or manual editing of the code is not a good practice 
since the likelihood of making mistakes increases exponentially.  Automation of the changes is always a favorable 
practice.  
 
This macro first finds the list of programs that need modification in the input directory.  It then automatically searches 
for the code that needs to be replaced and modifies each program as needed.  The modified programs will then be 
saved in the specified output directory. 

METHODOLOGY 
This macro first obtains a list of programs that need modification.  The macro will then loop through the list of 
programs, replacing the old code with the new code for each program.  Finally, the modified programs will be saved 
into a separate directory. 

MACRO CALL AND PARAMETER 
The following code is a sample macro call: 
 

%replace(drive_letter=r, 
         pgm_in_path=\Programs\Pgm in, 
         pgm_out_path=\Programs\Pgm out, 
         replace_code=%nrstr(%let macro_var=Replace This Code;), 
         new_code=%nrstr(%let macro_var=This is the New Code;)); 

 
The drive_letter parameter is the Windows drive letter for the input and output directories.  The parameters for the 
input and output directories are pgm_in_path and pgm_out_path, respectively.  The parameters for the old and new 
code are replace_code and new_code, respectively.  If the old or the new code has a macro token in it, it must be 
quoted using the %NRSTR function so that the macro variable or statement is not resolved. 

CREATE LIST OF PROGRAMS TO BE MODIFIED 
Before running the macro, all programs that need modification should be put in a directory that only contains those 
programs.  This will be the input directory for the macro.  The macro will first create a list of the programs in the input 
directory.  This list will be written to a text file in the input directory.  The following code will create the list of input files 

in Windows: 
 

options noXwait; 
 
X "&drive_letter:"; 
X "cd &drive_letter:&pgm_in_path"; 
X " dir  /b/s .\*.sas > pgms.txt "; 

 
The noXwait option closes the DOS command window when the DOS commands finish running.  After these 
commands run, there is a file called pgms.txt in the input directory that looks like this: 
 

r:\Programs\Pgm in\Program1.sas 
… 
r:\Programs\Pgm in\Program10.sas 

 
Next, the length of the longest program name and path in the text file is determined.  The TRUNCOVER option is 
used to keep the INPUT statement from moving to the next line because of a shorter than expected line in the input 
file.  It is needed if the input file has lines of varying lengths.  The logical record length option in the infile statement is 
set to 1000.  In Windows, the default is 256, which may be shorter than the length of the file name and path.  In the 
data step below, it is set to 1000 to make sure the entire file name is read: 
 

data _null_; 
  infile pgm length=inputlength lrecl=1000 end=endfile truncover; 
  input temp $; 
  retain maxlength; 
  if _n_ = 1 then maxlength = 0; 
  if inputlength > maxlength then maxlength = inputlength; 
  if endfile then call symput ('max_path_length', trim(left(put(maxlength, 5.)))); 
run; 

 
The text file with the input program names is then read into a data set: 
 

data pgm_in; 
  infile pgm lrecl=&max_path_length pad; 
  input full_pgm_path $char&max_path_length..; 
  pgm_name=reverse(scan(reverse(full_pgm_path),1,'\')); 
run; 

 
The program name is separated from the path by using the REVERSE function to get the name of the program before 
the first \ in the path.  The SCAN function then takes only the program name, and another call to the REVERSE 
function puts the program name back in the correct order.  The final part of this section of the macro is to put the list 
of programs into a macro variable separated by a delimiter, an asterisk in this case: 
 

proc sql noprint; 
  select pgm_name into: pgm_list separated by '*' 
    from pgm_in; 
quit; 

 

REPLACE CODE MACRO  
The next part of the macro is a subroutine that replaces code for one program.  First, file references are specified for 
the input and output programs: 
 

%macro replace_1pgm (file_in=); 
 
  filename in "&file_in"; 
  filename out "&drive_letter:&pgm_out_path\&file_in"; 

Next, the length of the longest line in the input program is found using the same data step that found the longest 
program name and path.  The maximum line length is put into the max_line_length macro variable.  Then the input 
program is read into a data set: 
 

  data pgm_in; 
    infile in pad lrecl=&max_line_length; 
    input pgm_line $char&max_line_length..; 
  run; 

 
The INDEX function is used to determine if the code to be replaced exists on a line.  If it does, the column that the 
code starts on is returned by the function.  Otherwise, the function returns 0.  The new version of the program is then 
created by using the TRANWRD function to replace the old code with the new code if it exists on a line: 
 

  data pgm_out; 
    set pgm_in; 
    replace_col = index(pgm_line, "&replace_code"); 
    if replace_col > 0 then  
       pgm_line = tranwrd(pgm_line, "&replace_code", "&new_code"); 
  run; 

 
Finally, the new program is written to the output directory: 
 

  data _null_; 
    set pgm_out; 
 file out lrecl=&max_line_length.; 
 put @1 pgm_line $char&max_line_length..; 
  run; 
 
%mend replace_1pgm; 

 
Back in the main macro, a macro DO WHILE loop is used to call the subroutine for each program in the input program 
list: 
 

%let i=1; 
%do %while(%scan(&pgm_list,&i,*) ne ); 
  %replace_1pgm (file_in=%scan(&pgm_list,&i,*));  
  %let i=%eval(&i+1); 
%end; 

 

DIFFERENCES FOR USE ON MVS 
The drive_letter parameter has no meaning on MVS.  The list of program names is obtained by using the SOURCE 
procedure and directing the output into a temporary file.  Using an ampersand enclosed in single quotes causes MVS 
to create temporary files.  The length of a MVS PDS library will not exceed 44 characters, and the member name will 
not exceed eight characters, resulting in a full file name length of 54.  The pgm_name variable is constructed as the 
PDSname(member) for use in the filename statement. 
 

filename tmp '&tmp'; 
filename pgm '&pgm'; 
 
proc source indd=origfile nodata outdd=tmp; 
 
data _null_; 
  file pgm lrecl=80; 
  infile tmp; 
  input @21 member $8.; 
  format fname $54.; 
  fname="&pgm_in_path"||"("||trim(member)||")"; 
  put @1 fname; 
run; 

TESTING 
After the changes have taken place each program is run separately and tested to ensure that the changes have taken 
place and that the new programs run without errors.  Scanning output for similarity is often the approach taken, but it 
is easy to overlook discrepancies.  Testing the equality of tables from the two solutions is a more accurate approach. 
 
The COMPARE procedure enables comparison of entire tables, columns or subsets of data. An example of PROC 
COMPARE is shown: 
 

proc compare data=dir1 out=dir2; 
run;    

 
Additionally, the Windows FC command and the MVS SUPERC utility can be used to compare files. 

CONCLUSION 
When dealing with a large number of programs, automation is essential to avoid errors due to typing.  Efficiency is 
also gained due to automation.  This technique works best when you have small changes that must be applied to a 
large number of programs.  This macro can also accommodate any number of frequent changes to a set of programs. 

